
#include <WiFi.h> 

#include <HTTPClient.h> 

#include <Wire.h> 

#include <Adafruit_BME280.h> 

 

// =============================== 

//   WLAN  (HIER ANPASSEN) 

// =============================== 

const char* WIFI_SSID = "hier WLAN-Name eintragen"; 

const char* WIFI_PASS = "hier WLAN-Passwort eintragen"; 

 

// ======================================= 

//   URL (HIER ANPASSEN) 

//   Google Apps Script Web-App URL: 

//   .../exec   (ohne Parameter) 

// ======================================= 

const char* GSCRIPT_URL = 

"Google Script-Link hier einfügen"; 

 

const uint32_t SAMPLE_INTERVAL_MS = 1000;  // 1 Hz messen 

const uint32_t SEND_INTERVAL_MS = 5000; 

 

uint32_t lastSample = 0; 

 

float minLux = NAN; 

float lastLux = NAN; 

float lastT = NAN; 

float lastRH = NAN; 

float lastHPA = NAN; 

 

// ===== Sensor ===== 

#define SDA_PIN 21 // ggf. Zahl ändern, sofern anderer Pin genutzt wird 

#define SCL_PIN 22 // ggf. Zahl ändern, sofern anderer Pin genutzt wird 

#define TSL_ADDR 0x39 

 

Adafruit_BME280 bme; 

 

// --- TSL2561 read 16-bit --- 

uint16_t rd16(byte reg) { 

  Wire.beginTransmission(TSL_ADDR); 

  Wire.write(0x80 | 0x20 | reg);  // Command + AutoIncrement + reg 

  Wire.endTransmission(false); 

  Wire.requestFrom(TSL_ADDR, (uint8_t)2); 

  if (Wire.available() < 2) return 0; 

  byte lo = Wire.read(); 

  byte hi = Wire.read(); 

  return (hi << 8) | lo; 

} 

 

// --- Lux calc (für Standard-Setup 402ms + 16x) --- 

float luxCalc(uint16_t ch0, uint16_t ch1) { 

  if (ch0 == 0) return 0.0f; 



  float r = (float)ch1 / (float)ch0; 

 

  if (r <= 0.5f)   return 0.0304f * ch0 - 0.062f * ch0 * powf(r, 1.4f); 

  if (r <= 0.61f)  return 0.0224f * ch0 - 0.031f * ch1; 

  if (r <= 0.80f)  return 0.0128f * ch0 - 0.0153f * ch1; 

  if (r <= 1.30f)  return 0.00146f * ch0 - 0.00112f * ch1; 

  return 0.0f; 

} 

 

void wifiConnect() { 

  WiFi.mode(WIFI_STA); 

  WiFi.begin(WIFI_SSID, WIFI_PASS); 

  Serial.print("WiFi connecting"); 

  while (WiFi.status() != WL_CONNECTED) { 

    delay(250); 

    Serial.print("."); 

  } 

  Serial.println(); 

  Serial.print("WiFi OK: "); 

  Serial.println(WiFi.localIP()); 

} 

 

bool sendToGoogleSheet(float lux, float t, float rh, float hpa) { 

  if (WiFi.status() != WL_CONNECTED) return false; 

 

  String url = String(GSCRIPT_URL) + 

               "?lux=" + String(lux, 1) + 

               "&temp=" + String(t, 1) + 

               "&rh=" + String(rh, 0) + 

               "&hpa=" + String(hpa, 1); 

 

  Serial.print("URL: "); 

  Serial.println(url); 

 

  HTTPClient http; 

  http.setFollowRedirects(HTTPC_STRICT_FOLLOW_REDIRECTS); 

  http.begin(url); 

 

  int code = http.GET(); 

  String resp = http.getString(); 

  http.end(); 

 

  Serial.print("GS HTTP "); 

  Serial.print(code); 

  Serial.print(" Resp: "); 

  Serial.println(resp); 

 

  return (code == 200); 

} 

 

uint32_t lastSend = 0; 

 



void setup() { 

  Serial.begin(115200); 

  delay(200); 

 

  Wire.begin(SDA_PIN, SCL_PIN); 

 

  // BME280 

  if (!bme.begin(0x76) && !bme.begin(0x77)) { 

    Serial.println("BME FAIL"); 

  } 

 

  // TSL2561 Power ON 

  Wire.beginTransmission(TSL_ADDR); 

  Wire.write(0x80);  // Control reg 

  Wire.write(0x03);  // Power ON 

  Wire.endTransmission(); 

 

  // TSL2561 Timing: 402ms + 16x 

  Wire.beginTransmission(TSL_ADDR); 

  Wire.write(0x81);        // Timing reg 

  Wire.write(0x00); 

  Wire.endTransmission(); 

 

  delay(200); 

 

  wifiConnect(); 

} 

 

void loop() { 

  uint32_t now = millis(); 

 

  // 1 Hz Messung 

  if (now - lastSample >= SAMPLE_INTERVAL_MS) { 

    uint16_t ch0 = rd16(0x0C); 

    uint16_t ch1 = rd16(0x0E); 

 

    lastLux = luxCalc(ch0, ch1) * 469.5f; 

    lastT   = bme.readTemperature(); 

    lastRH  = bme.readHumidity(); 

    lastHPA = bme.readPressure() / 100.0f; 

 

    // Minimum Lux seit letztem Send sammeln 

    if (isnan(minLux) || lastLux < minLux) { 

      minLux = lastLux; 

    } 

 

    Serial.print("  Lux="); Serial.print(lastLux, 1); 

    Serial.print(" ch0="); Serial.print(ch0); 

    Serial.print("  (min="); Serial.print(minLux, 1); Serial.print(")"); 

    Serial.print("  T="); Serial.print(lastT, 1); 

    Serial.print("  RH="); Serial.print(lastRH, 0); 

    Serial.print("  hPa="); Serial.println(lastHPA, 1); 



 

    lastSample = now; 

  } 

 

  // Senden im bisherigen Intervall, aber Lux = Minimum seit letztem Send 

  if (now - lastSend >= SEND_INTERVAL_MS) { 

    if (WiFi.status() != WL_CONNECTED) wifiConnect(); 

 

    // falls noch keine Messung passiert ist, minLux aus lastLux übernehmen (oder 0) 

    float luxToSend = isnan(minLux) ? (isnan(lastLux) ? 0.0f : lastLux) : minLux; 

 

    bool ok = sendToGoogleSheet(luxToSend, lastT, lastRH, lastHPA); 

    Serial.println(ok ? "Send OK" : "Send FAIL"); 

 

    // Reset für nächstes Fenster 

    minLux = NAN; 

 

    lastSend = now; 

  } 

 

  delay(10); 

} 

 

 


